



**LB\_3.1: Kennlinien des Sperrschicht-FET 2N 3819**




Bild LB\_3.1\_1: Schaltung zum Lehrbeispiel 3.1

\*\*\* Netzliste LB\_3.1 \*\*\*

```
V_UGS      G      0      -1V
V_UDS      D      0      10V
J_V1       D      G      0      J2N3819
.DC         LIN   V_UDS    0      24      1m
+ V_UGS    LIST  -2      -1.5  -1      -0.5  0
.TEMP      27
.PROBE
.END
```

Abschn.: **3.2.1**

**DC-Analyse**

AC-Analyse

Tran.-Analyse

**DC-Main-Sweep**

**DC-Nested-Sweep**

AC-Sweep

Param.-Sweep

Temp.-Sweep

**Lösung** siehe: PROBE-Fenster

*Trace* → *Add Trace*

Trace-Expression-Zeile: ID (J\_V1)

OK

Ende dieses Beispiels



**LB\_3.2: Kennlinien des MOS-FET IRF 150**




Bild LB\_3.2\_1: Schaltung zum Lehrbeispiel 3.2

\*\*\* Netzliste LB\_3.2 \*\*\*

```
V_UDS      1      0      10V
V_UGS      2      0      1V
M_V1       1      2      0      0      IRF150
.DC        LIN  V_UDS      0V      10V      1m
+ V_UGS    LIST  4      5      6
.TEMP      27
.PROBE
.END
```

Abschn.: **3.2.3**

**DC-Analyse**

AC-Analyse

Tran.-Analyse

**DC-Main-Sweep**

**DC-Nested-Sweep**

AC-Sweep

Param.-Sweep

Temp.-Sweep

**Lösung** siehe: PROBE-Fenster

*Trace* → *Add Trace*

Trace-Expression-Zeile: **ID (M\_V1)**

**OK**



**LB\_3.3: Modell-Parameter des Sperrschicht-FET 2N 4393**




Bild LB\_3.3\_1: Schaltung zum Lehrbeispiel 3.3

\*\*\* Netzliste LB\_3.3 \*\*\*

```
V_UDS      D      0      10V
V_UGS      G      0      -1V
J_V1       D      G      0      J2N4393-X1
J_V2       D      G      0      J2N4393-X2
J_V3       D      G      0      J2N4393-X3
.model     J2N4393-X1  NJF      (Beta=9.109m Betatce=-.5
+ Rd=1 Rs=1 Lambda=6m Vto=-1.422 Vtotc=-2.5m)
.model     J2N4393-X2  NJF      (Beta=9.109m Betatce=-.5
+ Rd=1 Rs=1 Lambda=6m Vto=-2.8   Vtotc=-2.5m)
.model     J2N4393-X3  NJF      (Beta=9.109m Betatce=-.5
+ Rd=1 Rs=1 Lambda=6m Vto=-0.7   Vtotc=-2.5m)
.DC        LIN  V_UGS    -3     0     1m
.TEMP      20
.PROBE
.END
```

Abschn.: **3.2.5**

**DC-Analyse**  
AC-Analyse  
Tran.-Analyse

**DC-Main-Sweep**  
DC-Nested-Sweep  
AC-Sweep  
Param.-Sweep  
Temp.-Sweep

**Lösung** siehe: PROBE-Fenster

Trace → Add Trace

Trace-Expression-Zeile: ID (J\_Vx)

OK

Ende dieses Beispiels



Prof. Dr.-Ing. Rainer Ose

Elektrotechnik  
für Ingenieure  
– Bauelemente mit PSpice –



Fachhochschule  
Braunschweig/Wolfenbüttel  
– University of Applied Sciences –

### LB\_3.4: Kleinsignalverstärker mit Sperrschicht-FET 2N 3819




Bild LB\_3.4\_1: Schaltung zur Arbeitspunktüberprüfung im Lehrbeispiel 3.4

\*\*\* Netzliste (1) LB\_3.4 \*\*\*

```
V_UDS      D      0      8V
V_UGS      G      0     -1.26V
J_V1       D      G      0      J2N3819
.DC        LIN   V_UDS    0      15V    1mV
.TEMP      27
.PROBE
.END
```

Abschn.: **3.2.6**

**DC-Analyse**

AC-Analyse

Tran.-Analyse

**DC-Main-Sweep**

DC-Nested-Sweep

AC-Sweep

Param.-Sweep

Temp.-Sweep

**Lösung** siehe: PROBE-Fenster

*Trace* → *Add Trace*

Trace-Expression-Zeile: **ID (J\_V1)**

**OK**

Fortsetzung: nächste Seite




Bild LB\_3.7\_2: Schaltung zur Kleinsignalverstärkung im Lehrbeispiel 3.4

\*\*\* Netzliste (2) LB\_3.4 \*\*\*

```

V_U1      ein 0  AC 15mV  SIN    0  15m   1kHz  0  0  0
V_UB      4  0  15V
J_V1      1  3  2  J2N3819
R_RD0     4  1  1500
R_RS0     2  0  300
R_R10     3  0  2MEG
R_Ra      aus 0  100k
C_C1      ein 3  10n
C_C2      1  aus 10n
C_C3      0  2  20u
.TRAN     0  5m 0  1u
.TEMP     27
.PROBE
.END
    
```

Abschn.: [3.2.6](#)

DC-Analyse  
AC-Analyse  
**Tran.-Analyse**

DC-Main-Sweep  
DC-Nested-Sweep  
AC-Sweep  
Param.-Sweep  
Temp.-Sweep

**Lösung** siehe: PROBE-Fenster

*Trace* → *Add Trace*

Trace-Expression-Zeile: V (ein) V (aus)

OK



**LB\_3.5: Kennlinienfelder des BJT BC 550 C**




Bild LB\_3.5\_1: Schaltung zum Lehrbeispiel 3.5

\*\*\* Netzliste (A–U) LB\_3.5 \*\*\*

```
V_UCE      C      0      1V
I_IB       0      B      DC    10uA
Q_V1      C      B      0      BC550C
.DC        LIN   V_UCE   0V    20V   1mV
+ LIN     I_IB   0      200uA  40uA
.TEMP     27
.PROBE
.END
```

\*\*\* Netzliste (E–I) LB\_3.5 \*\*\*

```
V_UCE      C      0      1V
I_IB       0      B      DC    10uA
Q_V1      C      B      0      BC550C
.DC        LIN   I_IB   0A    200uA   0.1uA
+ LIN     V_UCE  1V    15V   7V
.TEMP     27
.PROBE
.END
```

Abschn.: **3.3.3**

**DC-Analyse**  
AC-Analyse  
Tran.-Analyse

**DC-Main-Sweep**  
**DC-Nested-Sweep**  
AC-Sweep  
Param.-Sweep  
Temp.-Sweep

**Lösung** siehe: PROBE-Fenster

*Trace* → *Add Trace*

Trace-Expression-Zeile: IC (Q\_V1)

OK



**LB\_3.6: Hybridparameter des BJT BC 550 C**




Bild LB\_3.6\_1: Schaltung zur Bestimmung von h22 und h12 im Lehrbeispiel 3.6

\*\*\* Netzliste (A–U) LB\_3.6 \*\*\*

```

I_IC      0      C      SIN      40mA 5mA  1kHz  0      0      0
I_IB      0      B      DC       80uA
Q_V1     C      B      0       BC550C
.TRAN     0      5ms   0       .05u
.TEMP     27
.PROBE
.END
    
```

Abschn.: **3.3.4**

DC-Analyse

AC-Analyse

**Tran.-Analyse**

DC-Main-Sweep

DC-Nested-Sweep

AC-Sweep

Param.-Sweep

Temp.-Sweep

**Lösung** siehe: PROBE-Fenster

*Trace* → *Add Trace*

Trace-Expression-Zeile: V (C)

**OK**




Bild LB\_3.6\_2: Schaltung zur Bestimmung von  $h_{21}$  und  $h_{11}$  im Lehrbeispiel 3.6

\*\*\* Netzliste (E-I) LB\_3.6 \*\*\*

```

V_UBE    B    0    SIN    759mV    4mV    1kHz    0    0    0
V_UCE    C    0    8V
Q_V1     C    B    0    BC550C
.TRAN    0    5ms  0    .1u
.TEMP    27
.PROBE
.END
    
```

Abschn.: [3.3.4](#)

DC-Analyse

AC-Analyse

**Tran.-Analyse**

DC-Main-Sweep

DC-Nested-Sweep

AC-Sweep

Param.-Sweep

Temp.-Sweep

**Lösung** siehe: PROBE-Fenster

*Trace* → *Add Trace*

Trace-Expression-Zeile: IC (Q\_V1)

OK





**LB\_3.7:    Arbeitspunkteinstellung und -stabilisierung BC 550 C**




Bild LB\_3.7\_1: Schaltung zur Arbeitspunkteinstellung im Lehrbeispiel 3.7

\*\*\* Netzliste (1) LB\_3.7 \*\*\*

```
V_UB      3  0  20V
Q_V1      2  1  0  BC550C
R_RC      3  2  300
R_R1      1  3  22k
R_R2      0  1  950
.DC       LIN  TEMP 20 100  0.1
.PROBE
.END
```

Abschn.: **3.3.5**

**DC-Analyse**  
AC-Analyse  
Tran.-Analyse

**DC-Main-Sweep**  
DC-Nested-Sweep  
AC-Sweep  
Param.-Sweep  
Temp.-Sweep

**Lösung** siehe: PROBE-Fenster

*Trace* → *Add Trace*

Trace-Expression-Zeile: IC (Q\_V1)

**OK**




Bild LB\_3.7\_2: Schaltung zur Arbeitspunktstabilisierung im Lehrbeispiel 3.7

\*\*\* Netzliste (2) LB\_3.7 \*\*\*

```
V_UB      3  0  20V
Q_V1      2  1  4  BC550C
R_RC      3  2  250
R_R1      1  3  19.6k
R_R2      0  1  3450
R_RE      0  4  50
.DC       LIN  TEMP  20  100  0.1
.PROBE
.END
```

Abschn.: **3.3.6**

**DC-Analyse**

AC-Analyse

Tran.-Analyse

**DC-Main-Sweep**

DC-Nested-Sweep

AC-Sweep

Param.-Sweep

**Temp.-Sweep**

**Lösung** siehe: PROBE-Fenster

*Trace* → Add Trace

Trace-Expression-Zeile: IC (Q\_V1)

OK



**LB\_3.8: Modell-Parameter des BJT BC 550 C**




Bild LB\_3.8\_1: Schaltung zum Lehrbeispiel 3.8

\*\*\* Netzliste LB\_3.8 \*\*\*

```
V_UCE      C      0      1V
I_IB       0      B      DC      80uA
Q_V1      C      B      0      BC550C
.DC       LIN  V_UCE    0V    10V    1mV
.STEP     NPN  BC550C  (IS)  LIST  0.007f  7f    7p
.TEMP     27
.PROBE
.END
```

Abschn.: **3.3.7**

**DC-Analyse**

AC-Analyse

Tran.-Analyse

**DC-Main-Sweep**

DC-Nested-Sweep

**AC-Sweep**

Param.-Sweep

Temp.-Sweep

**Lösung** siehe: PROBE-Fenster

Trace → Add Trace

Trace-Expression-Zeile: IC (Q\_V1)

OK



**LB\_3.9: Transistfrequenz des Transistors BC 550 C**




Bild LB\_3.9\_1: Schaltung zum Lehrbeispiel 3.9

\*\*\* Netzliste LB\_3.9 \*\*\*

```
V_UCE      C      0      8V
I_IB       0      B      AC      12.5uA
+ SIN      80uA  12.5uA    1kHz  0      0      0
Q_V1      C      B      0      BC550C
.AC       DEC  1000  1k      1G
.TEMP     27
.PROBE
.END
```

Abschn.: **3.3.8**

DC-Analyse

**AC-Analyse**

Tran.-Analyse

DC-Main-Sweep

DC-Nested-Sweep

**AC-Sweep**

Param.-Sweep

Temp.-Sweep

**Lösung** siehe: PROBE-Fenster

Trace → Add Trace

Trace-Expression-Zeile: IC (Q\_V1) / IB (Q\_V1)

OK



**LB\_3.10: BJT als Schalter / Negator (BD 139)**




Bild LB\_3.10\_1: Schaltung zum Lehrbeispiel 3.10

\*\*\* Netzliste LB\_3.10 \*\*\*

```
V_Uq      2      0      1V
V_UB     3      0      5V
Q_V1     aus    1      0      BD139
R_Ri     2      1      {Rvar}
R_RC     aus    3      510
.PARAM   Rvar=5k
.DC      LIN   V_Uq  0V   5V   10mV
.STEP    PARAM  Rvar  LIST 1k   5k   10k   50k
.TEMP    27
.PROBE
.END
```

Abschn.: **3.3.9**

**DC-Analyse**  
AC-Analyse  
Tran.-Analyse

**DC-Main-Sweep**  
DC-Nested-Sweep  
AC-Sweep  
**Param.-Sweep**  
Temp.-Sweep

**Lösung** siehe: PROBE-Fenster

Trace → Add Trace

Trace-Expression-Zeile: V (aus)

OK



**LB\_3.11: Kennlinien von Thyristor und Triac**




Bild LB\_3.11\_1: Schaltung zur Aufnahme der Thyristor-Kennlinie im Lehrbeispiel 3.11

\*\*\* Netzliste (1) LB\_3.11 \*\*\*

```
V_Uq      1      0      30V
I_IG      0      2      DC      0mA
X_V1      A      2      0      2N1595
R_RV      1      A      50
.DC       LIN   V_Uq -70    60    10m
.TEMP     27
.PROBE
.END
```

Abschn.: **3.4.3**

**DC-Analyse**

AC-Analyse

Tran.-Analyse

**DC-Main-Sweep**

DC-Nested-Sweep

AC-Sweep

Param.-Sweep

Temp.-Sweep

**Lösung** siehe: PROBE-Fenster

*Trace* → *Add Trace*

Trace-Expression-Zeile: **I (R\_RV)**

**OK**




Bild LB\_3.11\_2: Schaltung zur Aufnahme der Triac-Kennlinie im Lehrbeispiel 3.11

\*\*\* Netzliste (2) LB\_3.11 \*\*\*

```
V_Uq      1      0      SIN      0      230V 1Hz      0      0      0
I_IG      0      2      DC       0mA
X_V1      0      2      AK       2N5444
R_RV      1      AK      10
.TRAN     0      1      0      1m
.TEMP     27
.PROBE
.END
```

Abschn.: **3.4.3**

DC-Analyse

AC-Analyse

**Tran.-Analyse**

DC-Main-Sweep

DC-Nested-Sweep

AC-Sweep

Param.-Sweep

Temp.-Sweep

**Lösung** siehe: PROBE-Fenster

*Trace* → *Add Trace*

Trace-Expression-Zeile: I(R\_RV)

OK